PRÁCTICA 7: Funciones de varias variables. Extremos

Derivadas parciales de una función

El primer objetivo será aprender a calcular derivadas parciales, gradientes, hessianos... de funciones de varias variables, y aplicar todo ello al cálculo de extremos relativos y condicionados.

Recordemos que el comando `diff` nos servía para calcular derivadas de funciones de una variable. Pues bien, también nos va a servir para calcular derivadas parciales de campos escalares de varias variables. Esto es

`diff(f(`$x_1, x_2\cdots$`),`$x_1, n_1, x_2, n_2, \cdots$`)`	derivada parcial de la función $f(x_1, x_2, \cdots)$ respecto a x_1, n_1 veces, respecto a x_2, n_2 veces,...

Calculemos como ejemplo $\frac{\partial^3 f}{\partial^2 x \partial y}$ y $\frac{\partial^2 f}{\partial x \partial y}$, donde $f(x, y) = \operatorname{sen}(x)\cos(y^2)$:

```
(%i1) f(x,y):= sin(x)*cos(y^2)$
(%i2) diff(f(x,y),x,2,y,1);
(%o2) 2 sin(x)y sin(y^2)
(%i3) diff(f(x,y),x,1,y,1);
(%o3) -2cos(x)y sin(y^2)
```

Con las derivadas parciales de primer orden construimos el vector gradiente ($\nabla f(a)$) de un campo escalar en un punto, y la matriz jacobiana ($Jf(a)$) de un campo vectorial en un punto. Para ambos cálculos utilizaremos el mismo comando: `jacobian(f,x)`. Por ejemplo: **jacobian**

```
(%i4) jacobian([f(x,y)],[x,y]);
(%o4) [cos(x) cos(y^2)   -2sin(x)y sin(y^2)]
```

hessian Ahora bien, si queremos calcular la matriz hessiana de un campo escalar, utilizaremos el comando `hessian` de la forma siguiente:

```
(%i6) hessian(f(x,y),[x,y]);
(%o6) [ -sin(x)cos(y^2)        -2cos(x)y sin(y^2)                    ]
      [ -2cos(x)y sin(y^2)     -2sin(x)sin(y^2)-4sin(x)y^2cos(y^2)   ]
```

Recuerda que las derivadas cruzadas de segundo orden coinciden para funciones "suficientemente buenas". Además, hay que comentar que, si estamos calculando el gradiente (o matriz jacobiana) de un campo escalar, mientras en el comando `hessian` la función se escribe $f(x, y)$, en el comando `jacobian` la función hay que escribirla entre corchetes. La razón es que este comando sirve tanto para campos escalares como vectoriales.

Extremos relativos

El método para encontrar extremos relativos de funciones de varias variables suficientemente derivables consiste en buscar primero los puntos críticos, es decir, puntos donde se anula el gradiente, y después estudiar la matriz hessiana en esos puntos. Los resultados que conocemos nos aseguran que todos los puntos extremos de una función están entre los puntos críticos, con lo que una vez calculados éstos nos dedicaremos a estudiar la matriz hessiana en ellos.

> Para comprobar qué tipo de puntos son los puntos críticos definimos la matriz hessiana, H.
> Si |H|>0 y $\partial_{x,x} f[x, y]$<0 en ese punto crítico hay un máximo.
> Si |H|>0 y $\partial_{x,x} f[x, y]$>0 en ese punto crítico hay un mínimo.
> Si |H|<0 en es punto crítico no hay máximo ni mínimo.
> Si |H|=0 puede existir o no extremo, en este caso haría falta realizar un estudio más detallado.

Ejemplo Calculemos los extremos relativos de la función $f(x,y) = x^3+3xy^2-15x-12y$.
Primero definimos la función, y calculamos su gradiente y su matriz hessiana.

```
(%i43)  f(x,y):= x^3+3*x*y^2-15*x-12*y;
        j:jacobian([f(x,y)],[x,y]);
        define(h(x,y),hessian(f(x,y),[x,y]));
```

(%o43) $f(x,y):=x^3+3xy^2+(-15)x+(-12)y$

(%o44) $\begin{bmatrix} 3x^2+3y^2-15 & 6xy-12 \end{bmatrix}$

(%o45) $\begin{bmatrix} 6x & 6y \\ 6y & -6x \end{bmatrix}$

Para calcular los puntos críticos podemos irnos a **Ecuaciones→Resolver sistema algebraico** e ir rellenando los datos que nos van pidiendo. De esta forma entra en acción el comando `algsys` para resolver sistemas de ecuaciones. A la hora de llamar a las ecuaciones, lo haremos de la siguiente forma:

```
(%i1) realonly:true;
(%o1) true
```

```
(%i46)  pcrit:algsys([j[1,1],j[1,2]],[x,y]);
(%o46)  [[x=2,y=1],[x=1,y=2],[x=-1,y=-2],[x=-2,y=-1]]
```

donde j[1,1] es el primer elemento del gradiente, es decir $\frac{\partial f}{\partial x}$ y j[1,2] es el segundo, es decir $\frac{\partial f}{\partial y}$. Observad que el resultado es una lista de listas de puntos; lista que además hemos llamado `pcrit` para luego poder acudir a ella a la hora de evaluar la matriz hessiana en los puntos críticos obtenidos.

Observad también que cómo únicamente nos interesan las raíces reales, hemos incluido la orden **realonly.** Calcularíamos ahora la matriz hessiana en cada uno de los puntos críticos para clasificarlos. Por ejemplo en el primero obtenemos:

```
(%i40) pcrit[1];
(%o40) [x=2,y=1]
```

(%i41) `h(x,y),pcrit[1];`

(%o41) $\begin{bmatrix} 12 & 6 \\ 6 & 12 \end{bmatrix}$

```
(%i42) determinant(%);
(%o42) 108
```

Por tanto, en el punto (2,1) la función *f* presenta un mínimo relativo. Así mismo, en (-2,-1) hay un máximo relativo:

```
(%i43) pcrit[4];
(%o43) [x=-2,y=-1]

(%i44) h(x,y),pcrit[4];
```

(%o44) $\begin{bmatrix} -12 & -6 \\ -6 & -12 \end{bmatrix}$

```
(%i45) determinant(%);
(%o45) 108
```

Mientras que en los puntos (1,2) y (-1,-2) la función *f* presenta puntos de silla.

```
(%i49) pcrit[2];
(%o49) [x=1,y=2]

(%i50) h(x,y),pcrit[2];
```

(%o50) $\begin{bmatrix} 6 & 12 \\ 12 & 6 \end{bmatrix}$

```
(%i51) determinant(%);
(%o51) -108
```

La gráfica de la función nos da también una idea de los extremos:

```
(%i10) wxplot3d(x^3+3*x*y^2-15*x-12*y, [x,-3,3], [y,-3,3])$
```

(%t10)



Y las curvas de nivel, centradas en el punto crítico (2,1), nos muestran claramente la presencia en ese punto de un mínimo:

```
(%i13) wxplot3d(x^3+3*x*y^2-15*x-12*y, [x,1.5,2.5], [y,0.5,1.5],
   [gnuplot_preamble, "set pm3d at b"])$
```

(%t13)



Prueba a hacer lo mismo con el resto de los puntos críticos, y prueba también distintas gráficas de la función centradas en dichos puntos.

Ejemplo Calculemos los extremos relativos de $g(x,y) = (x^2+3y^2)\,e^{1-x^2-y^2}$. Comenzamos definiendo g:

```
(%i49)   g(x,y):=(x^2+3*y^2)*exp(1-x^2-y^2)$
```

Calculamos ahora los puntos críticos como en el ejemplo anterior y tenemos:

```
(%i50)   j:jacobian([g(x,y)],[x,y])$
(%i51)   pcrit:algsys([j[1,1],j[1,2]],[x,y]);
(%o51)   [[x=0,y=0],[x=-1,y=0],[x=1,y=0],[x=0,y=-1],[x=0,y=1]]
```

y vamos a calcular la matriz hessiana y a clasificarla en cada uno de los puntos críticos encontrados:

```
(%i56) pcrit[1];
(%o56) [x=0,y=0]

(%i57) h(x,y),pcrit[1];
```

(%o57) $\begin{bmatrix} 2\,\%e & 0 \\ 0 & 6\,\%e \end{bmatrix}$

```
(%i58) determinant(%);
```

(%o58) $12\,\%e^2$

Así, en el punto $(0,0)$ tenemos un mínimo y, razonando de la misma forma en los otros cuatro puntos, dos puntos de máximo en $(0,-1)$ y $(0,1)$, así como dos puntos de silla en $(-1,0)$ y $(1,0)$. Podemos observar esto en la gráfica de la función, mejor aún si a la vez dibujamos las curvas de nivel:

```
(%i6) wxplot3d((x^2+3*y^2)*exp(1-x^2-y^2),[x,-2,2],[y,-2,2],
      [plot_format,gnuplot],[gnuplot_preamble,"set pm3d at b"])$
```

(%t6)



O la propia gráfica de la función.

```
(%i7) wxplot3d((x^2+3*y^2)*exp(1-x^2-y^2),[x,-2,2],[y,-2,2],
      [plot_format,gnuplot])$
```

(%t7)



Extremos condicionados

Usaremos el método de los multiplicadores de Lagrange para calcular extremos condicionados. Se trata de optimizar una función de varias variables en el conjunto de puntos que verifique una cierta ecuación o ecuaciones. Dada una función suficientemente diferenciable, y una curva o superficie en forma implícita, el método consiste en encontrar los puntos de dicha curva o superficie que verifica un sistema auxiliar, llamado "sistema de Lagrange", que dará el equivalente a los puntos críticos. En vez de comprobar la condición suficiente basada en la diferencial segunda, analizaremos la naturaleza de los puntos críticos, con razonamientos de tipo geométrico.
Por comodidad usaremos el símbolo *a* para el multiplicador.

Ejemplo. Hallar las distancias máxima y mínima del origen a los puntos de la elipse x^2+2*y^2=1**.**

Comenzamos definiendo la función *f*, la condición y la función auxiliar de Lagrange:

```
(%i1) f(x,y):=(x^2+y^2)^(1/2);
```
(%o1) $f(x,y):=(x^2+y^2)^{1/2}$

```
(%i2) g(x,y):=x^2+2*y^2-1;
```
(%o2) $g(x,y):=x^2+2\,y^2-1$

```
(%i3) F(x,y,a):=f(x,y)+a*g(x,y);
```
(%o3) $F(x,y,a):=f(x,y)+a\,g(x,y)$

Definimos el gradiente y planteamos el sistema de Lagrange:

```
(%i4) j:jacobian([F(x,y,a)],[x,y,a]);
```
(%o4) $\begin{bmatrix} \frac{x}{\sqrt{y^2+x^2}}+2\,a\,x & \frac{y}{\sqrt{y^2+x^2}}+4\,a\,y & 2\,y^2+x^2-1 \end{bmatrix}$

```
(%i5) pcrit:algsys([j[1,1], j[1,2], j[1,3]], [x,y,a]);
```
(%o5) $[[x=1,y=0,a=-\frac{1}{2}],[x=-1,y=0,a=-\frac{1}{2}],[x=0,y=\frac{1}{\sqrt{2}},a=-\frac{1}{2^{3/2}}],[x=0,y=-\frac{1}{\sqrt{2}},a=-\frac{1}{2^{3/2}}]]$

Por último, valoramos la función en los puntos críticos.

```
(%i6) f(x,y),pcrit[1];
```
(%o6) 1

```
(%i7) f(x,y),pcrit[3];
```
(%o7) $\frac{1}{\sqrt{2}}$

Por consideraciones geométricas llegamos a la conclusión de que las distancias máxima y mínima son, respectivamente, 1 y $\frac{1}{\sqrt{2}}$

EJERCICIOS PROPUESTOS

1)

a) Representar gráficamente y hallar los máximos y mínimos relativos de la función:

$$f(x,y) = x^4 + 9y^2 - 3x^2 - 6xy$$

b) Representar gráficamente las curvas de nivel en un entorno de los máximos o mínimos hallados, para comprobar los resultados obtenidos.

2)

a) Representar gráficamente y hallar los máximos y mínimos relativos de la función:

$$f(x,y) = x^5 - 8x^3 + 25x^2 - 32x - 6xy + 9y^2$$

b) Representar gráficamente las curvas de nivel en un entorno de los máximos o mínimos hallados, para comprobar los resultados obtenidos.

3)

a) Representar gráficamente y hallar los máximos y mínimos relativos de la función:

$$f(x,y) = x^4 + y^4 + 2x^2y^2$$

b) Representar gráficamente las curvas de nivel en un entorno de los máximos o mínimos hallados, para comprobar los resultados obtenidos.

4) Representar gráficamente la elipse de ecuación

$$13x^2 - 10xy + 13y^2 - 72 = 0$$

Hallar las distancias máxima y mínima del origen a dicha elipse.